

3.5 プログラミング proc(n)

ver. 0.2

前回のprocのところで扱ったプログラムのモジュール化は、リターンが1変数(行列としての1変数)のもっとも使われるケースですが特殊なケースでした。移動幾何平均のprocであるmgav(m,t)について、その仕組みを図解すると下の図のようになります。

return		input
n	proc mgav	m, t
	<u>local</u> :n,i,j,k	
	計算	

この場合、return戻り値はnだけの1つです。GAUSSでは、この1個のreturnがprocの外部のプログラムで用いられたり、ほかのprocのinputに引き渡されたりします。そういう意味で、しばしばプログラムでは

```
proc(1)=mgav(m,t);
  local n,i,j,k;
  .....
  retp(n);
endp;
```

のように書かれることもあります。この(1)というのはreturnの数が1個という意味です。なお、returnが2個以上の関数は、そのreturnの数だけの数字をここに書きます。もし、returnがない場合には(0)を書きます。それぞれのreturnの数のケースについて例をあげて説明していきましょう。

リターンが0個の場合

ここでいうreturnの数が0個というのは、動作をしないまたは画面表示をしないという意味ではなくて、procの外部に受け渡す変数がないという意味です。主に、画面表示やフォーマットに関することだけをするにに使われます。条件分岐ifのところでとりあげたt値による棄却決定のプログラムをprocedureに変更した上で、この0リターンのケースとして扱ってみましょう。

プロシジャー

```
proc(0)=tdecision(t,df,pp);
  local p;
  p=2*cdftc(t,df);
  if p<=pp;
    print /rz "Reject H0: beta=0 at" pp*100 "% level";
```

```

else;
    print /rz "Do NOT reject H0 at" pp*100 "% level";
endif;
print " |t|<" p;
endp;

```

上のような、どんなプログラムにもあとで繰り返し利用できるprocedureが完成しました。見てわかるように、ret p文がありません。外部にreturnとして受け渡す変数がまったくないということはこのことは意味します。なお、このプログラムだけでは動きません。procの外部で引数のインプットに何か数値を与えなければなりません。一番簡単な方法は、おのこの引数に直接数値を代入して、

```
call tdecision(1.94,4,0.1);
```

とそのprocedureの外部に書いてやれば、以前と同じように次のように画面表示します。

画面表示

```

Do NOT reject H0 at          10 % level
|t|<          0.12437736

```

このように、proc(0)のリターンの数が0個というのは、print文などでいくつの行にわたって出力されようとリターンの数には関係がまったくありません。このret p文がないことを、上の場合について図で表現すると以下ようになります。

return	input
(なし)	t ,df,pp
proc(0)=tdecision	
<u>local:</u> p	
計算	
内部で画面表示	

ただし、この0個のリターンのprocの呼び出し方は、call文で呼び出すか、またはそのままその命令文を書くかのいずれかでなければなりません。1個のリターンと同じように何か別の変数に代入したり、print文の内容にしていけません。なぜなら、その関数自体にリターンがないと設定したからです。

リターンが1個の場合

これは、前回扱ったrndx2、mav、mgavの3つともにあてはまります。冒頭に書きましてのように、いずれの関数もproc(1)=という形で書きなおしても同じことになります。以前やったように、d:ドライブのdatafile1.txtというデータファイルを読みこんで、OLS estimator

である b を求めるプログラムをリターン 1 個の procedure 形式になおしてみましょう。行列方式で OLS を求めるには、

$$b = (X'X)^{-1}X'Y$$

$$Y = \begin{pmatrix} y_1 \\ y_2 \\ y_3 \\ \vdots \\ y_{27} \\ y_{28} \end{pmatrix}, X = \begin{pmatrix} 1 & x_1 \\ 1 & x_2 \\ 1 & x_3 \\ \vdots & \vdots \\ 1 & x_{27} \\ 1 & x_{28} \end{pmatrix}$$

の関係をいいます。 y 切片がある場合には、 x のデータの 1 列目をすべて 1 に設定する必要があります。 x が多変数になっても同様にして切片と係数の推定ができます。なお、この場合できる b は $()^{-1}$ の部分が 2×2 、そのあととあわせて $(2 \times 2) \times (2 \times 28) \times (28 \times 1)$ ですから、 2×1 になるはずで、 b の第 1 行目は切片の推定値、その 2 行目は X の係数の推定値になるはずで、

プロシジャー

```
proc(1)=lse(y,x);
  local b;
  b=inv(x'x)*x'y;
  retp(b);
endp;
```

同様に、

```
proc lse(y,x);
  local b;
  b=inv(x'x)*x'y;
  retp(b);
endp;
```

と書いても同じことになります。上のプログラムは procedure 本体だけですから、この引数インプットに数値または行列を与えてやった上で、戻り値を操作するなり画面表示するなりしないと何も動作はしません。いま、d:ドライブの datafile1.txt というデータを読み込んで、その 2 行目から 29 行目までのデータを用いて以前やったように、regression をしてみます。データの 1 列目を y 、2 列目を x として、さらに x には切片に対応する 1 からなる列ベクトルを水平方向につけて扱います。

プログラム

```

new; cls;
load data[29,2]=d:datafile1.txt;
y=data[2:29,1]; x=ones(28,1)~data[2:29,2];
bhat=lse(y,x);
print "coefficients:" bhat';

```

```

proc(1)=lse(y,x);
    local b;
    b=inv(x'x)*x'y;
    retp(b);
endp;

```

画面表示

```

coefficients:      77.795198      52.009829

```

上のプログラムでは、後半のprocedure lse(y,x)を呼び出すのに、xとyに数値を代入したあとの関数lse(y,x)の戻り値の値をbhatとにおいて、それをcoefficients:というメッセージつきで画面表示させています。戻り値は、procedureの中ではbという変数で扱われて、それが戻り値としてリターンされますが、このように名前の違う変数で受けることも可能です。ただし、bというローカル変数として宣言した値を直接扱ったり画面表示させるためのprint文の内容にはなることはできません。すなわち、

プログラム（間違った例）

```

new; cls;
load data[29,2]=d:datafile1.txt;
y=data[2:29,1]; x=ones(28,1)~data[2:29,2];
print b;
.....

```

ただし、b=lse(y,x);と関数の戻り値を一度bという変数に受けて代入すれば、print b;で画面表示できます。という意味で、ローカル変数で宣言された変数は、それがリターンであろうとなかろうと、外部から直接扱うことはできません。このところは、誤解のないようにしてください。もっとも、直接関数をそこに置くことによって、lse(y,x);とすれば、その戻り値、すなわちここではbの内容2行1列で画面表示されます。print lse(y,x);とした場合と同じ結果が得られます。このように、リターンが1個の場合の呼び出し方は、何かの変数に代入すること、または、そのステップを省略して画面表示させるためprint文の内容にすることができます。もしcall文でリターンが0個のように呼び出すと、リターンの変数の引渡しはされません。procの内部に外部への変数引渡しとは別に画面表示やその他のシステム制御に関する命令があれば、その部分だけを実行します。なお、関数そのものだけを書いてやると、その関数の引数の内容が画面表示されます。call文の役割としては、proc

の内容だけを実行させるかわりに、戻り値を表示したくない場合によく用いられます。

ポイント `call` プロシジャー名; `proc`の内容だけを実行して戻り値を表示しない

リターンの値が1個の場合について、`call`で`proc`を呼ぶと、その内容だけを実行して戻り値を表示しない例を示してみましょう。係数の推定とは別に、`proc`内部でresidual plotをする部分を`lse(y,x)`に付け加えてみます。ここでresidual残差には以下の関係を用います。この残差は、 Y と同じディメンションの 28×1 の列ベクトルになるはずです。

$$\hat{u} = Y - \hat{Y} = Y - Xb$$

$$\text{ここで、 } b = (X'X)^{-1} X'Y$$

プロシジャー

```
proc(1)=lse(y,x);
    local b,uhat,xaxis;
    b=inv(x'x)*x'y;
    uhat=y-x*b;
    xaxis=seqa(1,1,rows(x));
    uhat=uhat~zeros(rows(x),1);
    library pgraph;
    graphset;
    xy(xaxis,uhat);
    retp(b);
endp;
```

プログラム

```
new; cls;
load data[29,2]=d:datafile1.txt;
y=data[2:29,1]; x=ones(28,1)~data[2:29,2];
call lse(y,x);
```


```
proc(1)=lse(y,x);
    local b,uhat,xaxis;
    b=inv(x'x)*x'y;
    uhat=y-x*b;
    xaxis=seqa(1,1,rows(x));
    uhat=uhat~zeros(rows(x),1);
    library pgraph;
```

```

graphset;
xy(xaxis,uhat);
retp(b);
endp;

```

グラフ表示



上のように、call文で1リターンのprocを呼び出しているので、その戻り値はプログラム上では何もせずに、procの内部のグラフを描くところが実行されています。太字でしめされている前のプログラムにさらに付け加えられた部分では、まず、 $\text{uhat} = y - x * b$;でデータyとその推定値との差、すなわちresidual残差を求めています。これは、xとyの行数と同じ28行1列の列ベクトルです。これをx y グラフにプロットするわけですが、その準備に、x軸として1から28までの数列xaxisを用意します。一方、y軸にはこの残差uhatにすべてが0ばかりのベクトルを水平方向にマージしたものをuhatとしてやります。グラフを描く時の手順であるlibrary pgraph;でpgraphグラフライブラリを呼んでやって、graphset;でそのグローバル変数を初期化しています。x y グラフの関数に、準備したxaxisとuhatを射れてやれば、実際の残差のプロットを結んだ線と0の基準線が描けます。これは、x y グラフのグローバル変数のデフォルトがプロットを結ぶように設定されているためです。そのほかのプログラムの部分は、以前の係数だけを求めるprocと同じ作りになっています。ただし、localのあとのローカル変数の宣言では新たに加えた変数uhatとxaxisを書かないと、Undefined symbolとエラーが出てしまいますので注意が必要です。

リターンが2個の場合

例えば、リターンが2個のprocの作り方の要領は、proc(2) = としてやって、retp文の丸括弧の中にカンマで区切った2つの変数を入れるだけで、そのほかの部分はprocの組み立て方とまったく同じです。引き続きregressionのデータとプログラムを用いて、それを拡張していきます。まずは、リターンが2個の場合、係数のほかに、推定式全体のstandard error (なお、これは係数それぞれのseとは別物ですのでご注意ください)を求めてあわせて表示してみます。

プロシジャー

```
proc(2)=lse(y,x);  
    local b,uhat,n,k,s2hat,shat;  
    b=inv(x'x)*x'y;  
    uhat=y-x*b;  
    n=rows(x);  
    k=cols(x);  
    s2hat=uhat'uhat/(n-k);  
    shat=sqrt(s2hat);  
    retp(b,shat);  
endp;
```

プログラム

```
new; cls;  
load data[29,2]=d:datafile1.txt;  
y=data[2:29,1]; x=ones(28,1)~data[2:29,2];  
{b,shat}=lse(y,x);  
print "coefficients:" b';  
print "Std error of est:      " shat;
```

```
proc(2)=lse(y,x);  
    local b,uhat,n,k,s2hat,shat;  
    b=inv(x'x)*x'y;  
    uhat=y-x*b;  
    n=rows(x);  
    k=cols(x);  
    s2hat=uhat'uhat/(n-k);  
    shat=sqrt(s2hat);  
    retp(b,shat);  
endp;
```

画面表示

coefficients: 77.795198 52.009829

Std error of est: 52.331001

上のプログラムで、proc部分では、これまでのプログラムに加えて、データ行列 x の行数を n 、その列数を k とおいて、 $s2hat$ を $uhat'uhat/(n-k)$ で求めます。ここで $uhat'uhat$ の部分は、 $(1 \times 28) \times (28 \times 1)$ ですから 1×1 のスケーラーになっています。これをスケーラーである $n-k$ で割っています。ちょうどこれは行列表示の

$$\hat{\sigma}^2 = \frac{\hat{u}'\hat{u}}{n-k}$$

の計算に相当する部分です。この平方根をとって、Estimated regression standard error を求めています。shat にその値を代入して戻り値の 2 つ目の値としています。なお、よくこの $(n-k)$ の括弧を忘れてしまうことがあるので、基本的なことですが毎回確認作業をお忘れなく。GAUSS でも数学とおなじように掛け算と割り算は優先されます。前のプログラム同様、加えたローカル変数である $n, k, s2hat, shat$ を忘れずに宣言することと、proc の後の数字がリターンが 2 個ということで 2 にすると同時に、retp の丸括弧の中に b と $shat$ の 2 つの変数を入れます。これでリターン 1 個から 2 個への変更は完了しました。プログラムでは、この proc を前方参照して、 b と $shat$ の 2 つを画面表示させています。最大の関心事は 2 つ以上の戻り値の場合どうするかということです。「変数 = 」の形では受けられないことは想像にかたくありません。行列の関数のところで扱った 2 変数以上にわかれて関数の値がでてくる場合と同じように、{ 変数名A, 変数名B }= の形で受けます。すなわち、行列のところで扱った 2 つ以上の値がでてくる関数も実はこの proc の 2 以上のリターンの構造でできていたのです。3 つ以上のリターンになっても、{ 変数名A, 変数名B, 変数名C }= の形で受けて、計算結果である戻り値を proc の外部に取り出します。リターンが 1 個の場合と同じく、{ } 括弧内の変数名は proc から返される変数名と同じ必要は必ずしもありません。ただし、同じ数だけ変数は必要です。複数個 proc を連ねて変数を受け渡していく場合には同じ変数名であった方がプログラムがすっきりとして見やすくなります。なお、引数インプットとリターンとの関係は、この場合、下の図のようになります。

return		input
b, shat	proc lse	y, x
	<u>local:</u>	
	b, uhat, n, k,	
	s2hat, shat	
	計算	

リターンが多数の場合

上のプログラムをすすめて、基準統計量以外の、OLS から求められる基本的な情報を得

るプロトタイプとも言えるprocを作っておきましょう。推定式のstandard errorおよび、切片と係数のほかに、それぞれのstandard errorとそれに対応する t 値、R2を求めます。

プログラム

```
new; cls;
load data[29,2]=d:datafile1.txt;
y=data[2:29,1]; x=ones(28,1)~data[2:29,2];
{b,se,t,shat,df,r2}=lse(y,x);
print "y-intercept:" b[1] "          slope for x:" b[2];
print "          se:" se';
print "          t:" t';
print "Std of est : " shat;
print /rz "          df=" df;
print /rz "          R2=" r2;

proc(6)=lse(y,x);
    local b,uhat,n,k,df,s2hat,shat,varb,se,t,r2;
    b=inv(x'x)*x'y;
    uhat=y-x*b;
    n=rows(x);
    k=cols(x);
    df=n-k;
    s2hat=uhat'uhat/df;
    shat=sqrt(s2hat);
    varb=s2hat*inv(x'x);
    se=diag(sqrt(varb));
    t=b./se;
    r2=1-uhat'uhat/(y'(eye(n)-1/n*ones(n,1)*ones(n,1)')*y);
    retp(b,se,t,shat,df,r2);
endp;
```

画面表示

y-intercept:	77.795198	slope for x:	52.009829
se:	22.037400		3.8317273
t:	3.5301442		13.573468
Std of est :	52.331001		
df=	26		
R2=	0.87633124		

上のプログラムでは、まず後半のプロシジャのところでdfをn-kと定義してやって、n-kではなくてdfで割ることによってs2hatを計算しています。bについてのVarCov行列は、このs2hatと $(X'X)^{-1}$ をかけ合わせることによって求めます。これは、行列表示で

$$Var(b) = \hat{\sigma}^2 (X'X)^{-1}$$

に相当するものです。計算結果は2 × 2のディメンションになります。その対角要素が係数（正確には切片と係数の両方）のVarianceとなっています。

$$Var(b) = \begin{pmatrix} Var(b_1) & Cov(b_1, b_2) \\ Cov(b_2, b_1) & Var(b_2) \end{pmatrix}$$

したがって各係数のstandard errorを求めるには、上の計算結果のdiagonal対角成分の平方根をとることによって計算します。t-ratioは係数b / seの割合ですから、b割るseを要素対要素の計算で求めます。そういうわけで、b./seというぐあいにドットがついています。

$$t_i = \frac{b_i}{se_i}, \quad i = 1, 2$$

これは割る方の数seがベクトルであるためです。GAUSSでは、割る数やかける数がスケーラーの場合は、実際の数学と同じようにディメンションを考える必要もなく割ったりかけたり、行列のおおのの要素対スケーラーで“ / ”のマークで計算できます。しかしながら、割ったりかけたりする方の数がベクトルや行列の場合には、要素対要素の計算であることをしめすのにドットが必要です。（一部の言語には、さらにこれよりも厳しくスケーラーの割り算にまでドットを要求するものがあったり、その反対にまったくドットを要求しない言語も統計計量のソフトにはありますので注意が必要です。GAUSSは「要素対要素」の場合のみドットが必要で、「行列対スケーラー」の場合にはドットは不要です。）R2の計算では、行列表示形式では、

$$R2 = 1 - \frac{\hat{u}'\hat{u}}{y'(I_n - \frac{1}{n}ii')y}$$

を用いて計算しています。ここで

$$i = \begin{pmatrix} 1 \\ 1 \\ \vdots \\ \vdots \\ 1 \end{pmatrix} \quad \text{の } n \times 1 \text{ の列ベクトルです。}$$

もちろん、行列表示ではなくて、プログラムは複数行にまたがりますが、要素ごとに、

$$R2 = 1 - \frac{\sum_{t=1}^n \hat{u}_t^2}{\sum_{t=1}^n (y_t - \bar{y})^2}$$

を計算しても計算できます。同じことです。そのほかの部分と同じで、ローカル宣言のところに新たに加えたdf,varb,se,t,r2を書くことをわすれないでください。リターンとしてret pの丸括弧の中にさらにse,t,df,r2を追加して、これまでのbとshatの2つに加えて6つとなりますから、先頭のprocのところもproc(6)に変更します。

呼び出し方は、{ } 括弧の中にリターンとすべき変数5個すべてをカンマで区切って書いてイコールサインで関数と結びつけ、その関数の引数の中にデータまたは変数を代入すれば関数を呼び出せます。{b,se,t,shat,df,r2}=lse(y,x);のようになっているはずです。そのあとで、{ } 括弧の中そのおのおの変数をprint文で“ ”のコメント付で画面表示させています。なお、{ } 括弧の中にはprocのリターンの個数分変数を書かなくてはなりませんが、変数名は任意です。わかりやすいプログラムという点から慣習上、リターンと同じ変数名が使われることが多いようです。ただし、これらすべてのリターンをすべて利用したり画面表示する必要はありません。例えば、shatの画面表示行を消したり/* */でくるんだりしてコメント行として一時的に実行させないようにしても、何ら支障はありません。このように、公開されているprocの一部のリターン変数だけを利用してプログラムを組み立てることも可能になってきます。

最後に上のプログラムの構造は、おわかりのように、次のようになっています。

return		input
b,shat,	proc lse	y , x
df,se,t,r2	<u>local:</u>	
	b,uhat,n,k,	
	s2hat,shat,	
	varb,df,t,se,r2	
	計算	

ポイント proc(n)=関数名(引数,...);
 local 変数,...,変数;
 計算部分
 ret p(変数A, 変数B, 変数C,... nの個数だけ);
 endp;

ポイント {変数A, 変数B, 変数C,...nの個数だけ}=関数名()
 というふうに{ }括弧の中に変数を書いて複数リターンを呼び出す。